Singular perturbation expansion of the slow manifold of the Michaelis-Menten mechanism

Marc R. Roussel

This worksheet is intended to be studied alongside the notes for lecture 6.

The following procedure computes the *i*th coefficient of the expansion:

```
> gam := proc(i,s) option remember;
> if i=0 then s/(alpha*s+1-alpha);
> else (s*dgamds(i-1,s) - (alpha*s+beta*(1-alpha))*sum('dgamds(j,s)
  *gam(i-1-j,s)','j'=0..i-1))/(alpha*s+1-alpha);
> fi; end;
```

```
proc(i, s)
    option remember;
    if i = 0 then
        s / (alpha * s + 1 - alpha)
    else
        (s * dgamds(i - 1, s) - (alpha * s + beta * (1 - alpha)) * (sum('dgamds(j, s)
        * gam(i - 1 - j, s)', 'j' = 0 ..i - 1))) / (alpha * s + 1 - alpha)
    end if
end proc
```
(1)

The procedure above calls a procedure that computes the derivative of a coefficient with respect to *s*, provided below:

```
> dgamds := proc(i,s) option remember;
> diff(gam(i,s),s); end;
```

`proc(i, s) option remember; diff(gam(i, s), s) end proc` **(2)**

Note the use of option remember in both procedures above. This avoids unnecessary recomputation of coefficients.

Let's now look at some of those coefficients:

```
> gam(0,s);
```

$$\frac{s}{\alpha s + 1 - \alpha} \tag{3}$$

```
> factor(gam(1,s));
```

$$-\frac{s\left(-1+\alpha\right)^2\left(-1+\beta\right)}{\left(\alpha s + 1 - \alpha\right)^4} \tag{4}$$

```
> factor(gam(2,s));
```

$$\frac{s\left(-1+\alpha\right)^3\left(-1+\beta\right)\left(3\,\alpha\, s\,\beta + 2\,\beta\,\alpha - 4\,\alpha\, s - \alpha - 2\,\beta + 1\right)}{\left(\alpha s + 1 - \alpha\right)^7} \tag{5}$$

The following procedure computes the singular perturbation series for *c* to *n*th order in the singular perturbation coefficient μ.

```
> c := proc(n,s) sum('gam(i,s)*mu^i','i'=0..n) end;
```

`proc(n, s) sum('gam(i, s) * mu^i', 'i' = 0 ..n) end proc` **(6)**

```
> c(0,s);
```

(7)

$$\frac{s}{\alpha s + 1 - \alpha} \tag{7}$$

```
> c(1,s);
```

$$\frac{s}{\alpha s + 1 - \alpha} + \frac{1}{\alpha s + 1 - \alpha}\left(\left(s\left(\frac{1}{\alpha s + 1 - \alpha} - \frac{s\alpha}{(\alpha s + 1 - \alpha)^2}\right) - \frac{(\alpha s + \beta(1-\alpha))\left(\frac{1}{\alpha s + 1 - \alpha} - \frac{s\alpha}{(\alpha s + 1 - \alpha)^2}\right)s}{\alpha s + 1 - \alpha}\right)\mu\right) \tag{8}$$

Now let's see how the series converges for some specific values of the parameters:

```
> alpha := 0.5;
```

$$0.5 \tag{9}$$

```
> beta := 0.5;
```

$$0.5 \tag{10}$$

```
> mu := 0.2;
```

$$0.2 \tag{11}$$

```
> plot([c(0,s),c(1,s),c(2,s),c(3,s)],s=0..2,color=[red,green,blue,
  maroon]);
```


1.2
1
0.8
0.6
0.4
0.2
0
0
0.5
1
1.5
2